```
#include <WiFi.h>
#include <Wire.h>
#include <LiquidCrystal_I2C.h>
#include "DHT.h"
#include <ThingSpeak.h>

// ---------------- WIFI ----------------
const char* ssid = "Safii";
const char* password = "safiklaus";

// ---------------- THINGSPEAK ----------------
WiFiClient client;
unsigned long channelID = 3396333;
const char* writeAPIKey = "BXBJX5JDQ7UZU571";

// ---------------- WEB SERVER ----------------
WiFiServer server(80);

// ---------------- LCD ----------------
LiquidCrystal_I2C lcd(0x27, 16, 2);

// ---------------- PINS ----------------
#define DHTPIN 4
#define DHTTYPE DHT11

#define TRIG 5
#define ECHO 18

#define SOUND_PIN 34
#define PIEZO_PIN 35

#define BUZZER 23
#define LED_PIN 2

// ---------------- DHT ----------------
DHT dht(DHTPIN, DHTTYPE);

// ---------------- VARIABLES ----------------
float distance;
int soundValue;
int movementValue;
float temperature;
float humidity;

// ---------------- ULTRASONIC FUNCTION ----------------
float getDistance() {
  float total = 0;

  for (int i = 0; i < 5; i++) {
    digitalWrite(TRIG, LOW);
    delayMicroseconds(2);

    digitalWrite(TRIG, HIGH);
    delayMicroseconds(10);

    digitalWrite(TRIG, LOW);

    long duration = pulseIn(ECHO, HIGH, 30000);

    float d = duration * 0.034 / 2;

    total += d;

    delay(50);
  }

```

```
  return total / 5;
}

// ---------------- SETUP ----------------
void setup() {

  Serial.begin(115200);

  pinMode(TRIG, OUTPUT);
  pinMode(ECHO, INPUT);

  pinMode(SOUND_PIN, INPUT);
  pinMode(BUZZER, OUTPUT);
  pinMode(LED_PIN, OUTPUT);

  dht.begin();

  Wire.begin();

  lcd.init();
  lcd.backlight();

  lcd.setCursor(0, 0);
  lcd.print("Sleep Monitor");

  WiFi.begin(ssid, password);

  lcd.setCursor(0, 1);
  lcd.print("Connecting...");

  while (WiFi.status() != WL_CONNECTED) {
    delay(500);
    Serial.print(".");
  }

  Serial.println();
  Serial.println("WiFi Connected");

  lcd.clear();
  lcd.setCursor(0, 0);
  lcd.print("WiFi Connected");

  ThingSpeak.begin(client);

  server.begin();

  delay(2000);
  lcd.clear();
}

// ---------------- LOOP ----------------
void loop() {

  distance = getDistance();

  soundValue = analogRead(SOUND_PIN);

  movementValue = analogRead(PIEZO_PIN);

  temperature = dht.readTemperature();

  humidity = dht.readHumidity();

  bool abnormal = false;

  if (distance < 2 || distance > 80)
```

```
    abnormal = true;

  if (soundValue > 1500)
    abnormal = true;

  if (movementValue > 1200)
    abnormal = true;

  digitalWrite(BUZZER, abnormal);
  digitalWrite(LED_PIN, abnormal);

  lcd.clear();

  lcd.setCursor(0, 0);
  lcd.print("D:");
  lcd.print(distance, 0);

  lcd.setCursor(8, 0);
  lcd.print("S:");
  lcd.print(soundValue);

  lcd.setCursor(0, 1);
  lcd.print("T:");
  lcd.print(temperature, 1);

  lcd.setCursor(8, 1);
  lcd.print("M:");
  lcd.print(movementValue);

  ThingSpeak.setField(1, distance);
  ThingSpeak.setField(2, soundValue);
  ThingSpeak.setField(3, movementValue);
  ThingSpeak.setField(4, temperature);
  ThingSpeak.setField(5, humidity);

  ThingSpeak.writeFields(channelID, writeAPIKey);

  WiFiClient webClient = server.available();

  if (webClient) {

    webClient.println("HTTP/1.1 200 OK");
    webClient.println("Content-type:text/html");
    webClient.println();

    webClient.println("<html><head>");
    webClient.println("<meta http-equiv='refresh' content='2'>");
    webClient.println("<title>Sleep Monitor</title>");
    webClient.println("</head><body>");

    webClient.println("<h2>IoT Sleep Monitoring</h2>");
    webClient.println("<p>Distance: " + String(distance) + " cm</p>");
    webClient.println("<p>Sound: " + String(soundValue) + "</p>");
    webClient.println("<p>Movement: " + String(movementValue) + "</p>");
    webClient.println("<p>Temperature: " + String(temperature) + " C</p>");
    webClient.println("<p>Humidity: " + String(humidity) + " %</p>");

    if (abnormal)
      webClient.println("<h3 style='color:red;'>ALERT DETECTED</h3>");
    else
      webClient.println("<h3 style='color:green;'>NORMAL CONDITION</h3>");

    webClient.println("</body></html>");

    webClient.stop();
  }
```

```

  delay(2000);
}

```